Chris Troutner
8/22/07

Jupiter Waveamp Analysis

Intro
The over-current protection circuit in the Jupiter Waveamp is occasionally tripped when the board is connected to the test fixture. Preliminary tests show that increasing the inductance in the path to the load will cause an initial spike in current. We worry that the pogo pins in the test fixture are adding the inductance necessary to trip the current protection circuit.

The following pages take the test circuit and derive the corresponding equations of motion necessary to compare the accuracy of our model to measured values. Once the model is proven accurate, we can begin to play with the circuit elements in order to create a solution to this problem.



Rs	=	0.075 Ohm	Sense Resistor Ladder (Measured Value)
Z_pogo	=	?	Unknown Impedance of the pogo pins
R_rc	=	0.023 Ohm	Ribbon Cable Resistance (Measured Value)
L_rc	=	2 nH	Ribbon Cable Inductance (Specified Value)
C_rc	=	1 nF	Ribbon Cable Capacitance (Specified Value)
R_L	=	0.034 Ohm	Load Resistance (Specified Value)
C_L	=	0.24 uF	Load Capacitance (Specified Value)

Analysis

Assuming that the pogo pins only add inductance, the series inductances and resistances can be summed together into the following simplified circuit:



$$R = R_S + R_{RC} \approx 0.1\Omega$$

$$L = L_{RC} + L_{POGO} \geq 2nH$$

Looking closer at the impedance of the tail end:



$$Z = \frac{1}{sC_{RC}} \bigg\| \left(R + \frac{1}{sC_L}\right)$$

$$Z = \frac{1 + sRC_L}{s(RC_{RC}C_L + C_{RC} + C_L)}$$

Consider…

$$\frac{C_L}{C_{RC}} = 240 \qquad \frac{C_{RC}}{C_L} = 0.00417$$

Basically, C_{RC} is 0.4% of the size of C_L, so we may be able to neglect it. And since…

$$\lim_{C_{RC} \to 0} Z = \frac{1 + sRC_L}{sC_L} = R + \frac{1}{sC_L}$$

…which is the impedance if we removed C_{RC}, <u>we can neglect C_{RC}</u>.

Thus, the circuit further simplifies to:



This circuit has the corresponding transfer function for voltage and current:

$$H_1 = \frac{V_L}{V_I} = \frac{1 + sR_L C_L}{1 + sC_L(R_L + R) + s^2 C_L L}$$
$$H_2 = \frac{I}{V_I} = \frac{sC_L}{1 + sC_L(R_L + R) + s^2 C_L L}$$

These transfer functions are handy because we can take the inverse laplace transform of them, convolve the function with convolve the function with the arbitrary waveforms we recorded with the oscilloscope and compare how well our model matches reality.

But back to the problem at hand, we desire to control the overshoot (in-rush) current so that it doesn't't' trip the current protection circuit. The transfer function, H_2, is a second order equation, which fits the characteristic equation form:

$$H_2 = sC_L * \frac{\omega_n^2}{\omega_n^2 + 2s\zeta\omega_n + \omega_n^2}$$

$$\omega_n = \frac{1}{\sqrt{LC_L}} \qquad \zeta = \frac{R + R_L}{2L\omega_n}$$

Where ω_n is the circuits natural (resonant) frequency and ζ (zeta) is the dampening factor. Plugging in our component values, we see that:

$$\omega_n = 0.456 * 10^8 \text{ (rads/sec)} \qquad \zeta = 0.734$$

Summary

Since ζ is less than 1, we know the system is underdamped. In order to minimize the overshoot, we want to make ζ equal close to 1 – which would create a critically damped system. Looking at the equation for ζ, we see that the numerator is simply the sum of the systems resistances:

$$\zeta = \frac{R + R_L + R_c}{2L\omega_n} = 1$$

$$\Rightarrow R_c = 0.117\Omega$$

Thus, **adding a series resistance of 0.117 Ohms**, we can remove the overshoot and keep the over-current protection circuitry from tripping. The simulation profiles attached to this report show the results.

*Uncompensated System

Date/Time run: 08/22/07 18:00:17 Temperature: 27.0



A1:(611.366n,25.335) A2:(0.000,0.000) DIFF(A):(611.366n,25.335)

*Compensated System

Date/Time run: 08/22/07 17:57:38 Temperature: 27.0



A1:(611.366n,19.312) A2:(0.000,0.000) DIFF(A):(611.366n,19.312)

Date: August 22, 2007 Page 1 Time: 17:59:52

```
*Uncompensated System

*This approximation circuit is broken into four Sections:
*-Over-Current Sense Resistor
*-Pogo pin Impedence
*-Ribbon Cable
*-Load

*Chris Troutner
*8/22/07

*              V1 V2 Td Tr Tf Tw Per
Vs 1  0  PULSE(   0  45 .5u   .5u   .5u   10u   20u)


*Sense Resistor
Rs    1  3  .075

*Pogo Pin
Lp    3  4  2n

*Ribbon Cable
Rrc   4  5  .023
Lrc   5  6  2.25n
Crc   6  0  1n

*Load
Rl    6  7  .034
Cl    7  0  .24u

*Compensation Components
*Rc      2  3  .117

.TRAN .01u  3u 0  .01u

.PROBE

.END
```

```
*Compensated System

*This approximation circuit is broken into four Sections:
*-Over-Current Sense Resistor
*-Pogo pin Impedence
*-Ribbon Cable
*-Load

*Chris Troutner
*8/22/07

*              V1 V2 Td Tr Tf Tw Per
Vs 1  0  PULSE(   0  45 .5u   .5u   .5u   10u   20u)


*Sense Resistor
Rs    1  2  .075

*Pogo Pin
Lp    3  4  2n

*Ribbon Cable
Rrc   4  5  .023
Lrc   5  6  2.25n
Crc   6  0  1n

*Load
Rl    6  7  .034
Cl    7  0  .24u

*Compensation Components
Rc    2  3  .117

.TRAN .01u  3u 0  .01u

.PROBE

.END
```